
ODS and Office Integration 

ABSTRACT 
The growth of ODS destinations has provided many choices when it comes to getting your SAS® output into your 
favorite spreadsheet or word processor. Find out the various and best ways ODS integrates with Microsoft® Office 
and OpenOffice.org. Putting SAS output into documents and spreadsheets has never been so good or so easy. 

INTRODUCTION 
Today, I ordered a pizza to be delivered from Lucci's, one of my local pizza restaurants. I realized that the SAS 
Output Delivery System (ODS) is not unlike Lucci's. Lucci's has a good selection of pizzas to choose from, depending 
on my mood. If I have friends over, I can order several different types of pizza to satisfy everyone's needs. The 
number of output types that SAS ODS can deliver is as varied as the pizzas I can order from Lucci's. That's a good 
thing because the number of consumers that ODS delivers SAS output to is growing. 
 
The most visible consumers of ODS output are the office suites—Microsoft Office and OpenOffice.org—and more 
specifically, word processors, spreadsheets, databases, and Web browsers. ODS is capable of delivering output as 
simple as a cheese pizza, or as complex as the double-layered spinach pie I had for dinner. This paper shows the 
various ways that ODS can be used to deliver your SAS output to your customers, based on their needs. 
 
A short glossary with terminology that is relevant to this discussion is provided at the end of this paper. 

WHAT'S YOUR FAVORITE PIZZA? 
Most of us will eat any number of toppings on our pizza, but we all have our favorites and that's what we tend to eat. 
Our favorite might change from time to time, but what we like and don't like doesn't change that much. Choosing 
destinations for your ODS output is similar to choosing a pizza—there are one or two destinations that you always 
want, and a few others that you might select from time to time, and what you select sometimes depends on who you 
are ordering for—yourself, your family, or your friends (your customers). 
 
For example, if your customer wants word processing, then Rich Text Format (RTF) is probably your destination of 
choice. ODS RTF does a good job of getting your SAS output into your favorite word processor. With SAS® 9.2, the 
tagsets.rtf destination gives you even better RTF, as well as more control over the RTF it creates. 
 
Your customer might want a destination that is easily consumable by their favorite spreadsheet application, and if 
they aren't particular about how their spreadsheet looks, then data with comma-separated values (CSV) is an easy 
choice that doesn't require any more effort than it does to make a cheese pizza. If they want more toppings (so to 
speak), then, importing HTML might be a better choice. If they want a fully loaded spreadsheet with all the bells and 
whistles, then the tagsets.excelxp destination is your choice. 
 
There are many ODS destinations that are useful for publishing your reports to the Web; HTML is an obvious choice, 
and another choice is ODS PDF. ODS LaTeX is also a good choice when the final destination is PDF. When your 
target audience is using phones or PDAs, there is Compact HyperText Markup Language (cHTML) and Wireless 
Markup Language (WML). If you want to reach all of your potential audience, you might need all of these 
destinations. 
 
When it comes to databases, CSV files are still common, although Extensible Markup Language (XML) is becoming 
more popular. The ODS MARKUP family and the XML LIBNAME engine both provide XML destinations that can be 
easily consumed by the various databases. There is even an SQL tagset that creates INSERT statements from 
PROC PRINT. If you have customers that think Microsoft Excel is a database, then CSV or the tagsets.excelxp 
destinations are your best choices. 
 
ODS provides enough variety of output for anyone, and with the abilities of ODS MARKUP, the possibilities for more 
varieties are endless. For example, more and more applications are standardizing XML as a file format. As those 
standards solidify, ODS will be there with destinations to match them. 

EVERYONE LIKES PLAIN CHEESE PIZZA 
You can't complain about plain cheese pizza; it isn't anything special, but it doesn't surprise you either. There aren't 
any hidden ingredients, and what you get is very predictable, although very plain. SAS output in CSV format is the 
cheese pizza of ODS. 
 
CSV files have been used for data exchange for decades. They can be used and read by almost any application. The 
result is not pretty, but it is quite useful and predictable. PROC EXPORT can be used to create CSV files. However, 
ODS is much more flexible and does a better job of creating files that work well with Microsoft Excel or the 
OpenOffice.org Calc spreadsheet program. 
 
ODS CSV is a member of the ODS MARKUP family of destinations, which means it can be modified to work in any 
way that you desire. As is, the CSV destination comes with a number of options to control the way the output looks. 
 
The default behavior for the CSV destination is to print only the tabular data, to enclose strings in double quotation 
marks, and to use a comma as the field separator. There are two other CSV destinations that add additional 
behavior; the CSVBYLINE destination adds bylines to the output; the CSVALL destination adds bylines, titles, 
footnotes, and notes. These last two destinations are defined entirely by the CSV destination, but they change their 
behavior by automatically setting options. The CSV destination has more options than almost any other ODS 
destination. 
 
The OPTIONS option is unique to the ODS MARKUP family of destinations. Each markup destination is defined by a 
tagset, and tagsets can be created or modified by anyone. The options that a markup destination takes are malleable 
and can change or grow as we desire. That is good and bad. It's good because we can change our plain cheese 
pizza to any way we want. It's bad because we need a menu to know what we can choose. The good news is that the 
menu is built into the destination. Every ODS MARKUP destination that has the OPTIONS option also has a DOC 
option for documentation. All we need to do is ask it for help. Now, if we want to add fresh basil to our plain cheese 
pizza, we'll know how to order it. 

 
ods csv file="test.csv" options(doc="help"); 
 
NOTE: Writing CSV Body file: test.csv 
============================================================================== 
The CSV Tagset Help Text. 
  
This Tagset/Destination creates output in comma separated value format. 
  
Numbers, Currency, and percentages are correctly detected and show as numeric  
values. Dollar signs, commas, and percentages are stripped from numeric values by  
default. 
============================================================================== 

 
The options that are supported by this tagset are shown in the following sample code. 
 

ods csv options(doc='Quick'); 
  
ods csv options(currency_as_number='yes' percentage_as_number='yes'  
delimiter=';');  
  
Doc: No default value. 
     Help: Displays introductory text and options. 
     Quick: Displays available options. 
  
Delimiter:   Default Value ',' 
     Sets the delimiter for the values.  Comma is the default.  Semi-colon is 
     a popular setting for European sites. 
  
currency_as_number:   Default Value 'No' 
     If 'Yes' currency values will not be quoted. 
     The currency values are stripped of punctuation and currency symbols 
     so they can be used as a number. 

percentage_as_number:   Default Value 'No' 
     If 'Yes' percentage values will not be quoted. 
     The percentages are stripped of punctuation and the percent sign 
     so they can be used as a number. 
  
Currency_symbol:   Default Value '$' 
     Used for detection of currency formats and for  
     removing those symbols so Excel will like them. 
     Will be deprecated in a future release when it is 
     no longer needed.         
  
Decimal_separator:   Default Value '.' 
     The character used for the decimal point. 
     Will be deprecated in a future release when it is no longer needed. 
  
Thousands_separator:   Default Value ',' 
     The character used for indicating thousands in numeric values. 
     Used for removing those symbols from numerics so Excel will like them. 
     Will be deprecated in a future release when it is no longer needed. 
  
Quoted_columns:   Default Value '' 
     A list of column numbers that indicate which values should be quoted, 
     for example, Quoted_columns="123". 
  
Bylines:   Default Value: No  
     If yes bylines will be printed. 
  
Titles:   Default Value: No  
     If yes titles and footnotes will be printed. 
  
Notes:   Default Value: No 
     If yes Note, Warning, Error, and Fatal notes will be printed. 
  
Proc_Titles:   Default Value: No  
     If yes titles generated by the procedures will be printed. 
 

Here is an example that shows how to use the CSV destinations. 
 
ods csv file="csv.csv"; 
 
ods csvall file="csvall.csv"; 
 
ods tagsets.csvbyline file="csvbyline.csv"; 
 
ods csv(2) file="csvsemi.csv" options(Delimiter=';'); 
 
proc reg data=sashelp.class; 
   model Weight = Height Age; 
run;quit; 
 
ods _all_ close; 
 
 

In the previous code, the fourth ODS CSV statement uses the destination alias feature to allow two CSV destinations 
to be open at the same time. The second CSV destination sets the delimiter to a semicolon. Using the semicolon as a 
delimiter is common in Europe. 

The CSV files that are created by this program are predictable and plain. The biggest difference is that SAS titles and 
procedure titles are included by the CSV file created by the CSVALL destination. 

 
The SAS System 
 
The REG Procedure 
Model: MODEL1 
Dependent Variable: Weight 
"Number of Observations Read",19 
"Number of Observations Used",19 
 
"Analysis of Variance",,,,, 
"Source","DF","Sum ofSquares","MeanSquare","F Value","Pr > F" ) 
"Model",2,7215.63710,3607.81855,27.23,"<.0001" 
"Error",16,2120.09974,132.50623,, 
"Corrected Total",18,9335.73684,,, 
 
"Root MSE",11.51114,"R-Square",0.7729 
"Dependent Mean",100.02632,"Adj R-Sq",0.7445 
"Coeff Var",11.50811,, 
 
"Parameter Estimates",,,,, 
"Variable","DF","ParameterEstimate","StandardError","t Value","Pr > |t|" 
"Intercept","1",-141.22376,33.38309,-4.23,0.0006 
"Height","1",3.59703,0.90546,3.97,0.0011 
"Age","1",1.27839,3.11010,0.41,0.6865 
 

The output from the last CSV destination (shown on the previous page in the fourth ODS CSV statement) shows only 
the tables and has a semicolon instead of a comma as its delimiter. 

 
"Number of Observations Read";19 
"Number of Observations Used";19 
 
"Analysis of Variance";;;;; 
"Source";"DF";"Sum ofSquares";"MeanSquare";"F Value";"Pr > F"  
"Model";2;7215.63710;3607.81855;27.23;"<.0001" 
"Error";16;2120.09974;132.50623;; 
"Corrected Total";18;9335.73684;;; 
 
"Root MSE";11.51114;"R-Square";0.7729 
"Dependent Mean";100.02632;"Adj R-Sq";0.7445 
"Coeff Var";11.50811;; 
 
"Parameter Estimates";;;;; 
"Variable";"DF";"ParameterEstimate";"StandardError";"t Value";"Pr > |t|" 
"Intercept";"1";-141.22376;33.38309;-4.23;0.0006 
"Height";"1";3.59703;0.90546;3.97;0.0011 
"Age";"1";1.27839;3.11010;0.41;0.6865 
 

ONE PEPPERONI PIZZA TO GO PLEASE! 
CSV files might be the oldest and most simplistic way to exchange data, but HTML is also a good way. Many 
spreadsheets and word processors know how to import HTML. The result is not perfect, but it works fairly well and 
looks nicer than importing CSV files. 
 
The ODS MARKUP family supplies a number of HTML destinations. If your target is a Microsoft Office application, 
the MSOFFICE2K destination is your best choice. The MSOFFICE2K destination adds specific Microsoft directives to 
the HTML that help Microsoft Word and Microsoft Excel do a better job of importing. 
 
If your targets are phones and PDAs, then the cHTML or WML destinations are your best choices, although you 
should use the lesser-known IMODE destination if your customers are in Japan. 

For the Web, your best choice is usually the HTML destination, which is the HTML4 tagset. You might want to look at 
the PHTML or HTMLCSS tagsets if you are working in HTML and you don't like the output from the HTML4 tagset. 
 
If accessibility is a need, the HTML4 tagset is compliant with Section 508 of the U.S. Rehabilitation Act of 1973. 
Because this destination is a tagset, it can be modified for additional accessibility needs. 
 
Currently, there are no options with the HTML tagsets, but there are a number of custom HTML tagsets that have 
added features. They are available for download from the Web at support.sas.com/rnd/base/topics/odsmarkup/. 
 
Here is an example showing the various HTML destinations. 
 

ods imode file="imode.html"; 
 
ods chtml file="chtml.html"; 
 
ods phtml file="phtml.html"; 
 
ods htmlcss file="htmlcss.html"; 
 
ods html4 file="html4.html"; 
 
ods tagsets.xhtml file="xhtml.html"; 
 
ods tagsets.msoffice2k file="msoffice.html"; 
 
proc reg data=sashelp.class; 
   model Weight = Height Age; 
run;quit;   
  
ods _all_ close; 

 
The output from the IMODE destination is the ugliest of all because it does not support tables (Figure 1). 
 

 
Figure 1.  Output Using the Tagset-Based IMODE Destination 

The output from the cHTML destination looks good, considering how compact it is (Figure 2). 
 

 
Figure 2.  Output Using the Tagset-Based cHTML Destination 
 
 
The output from the PHTML destination (Figure 3) is not so pretty, but it works well if you just want simple style 
sheets. 
 

 
Figure 3.  Output Using the Tagset-Based PHTML Destination 

The output from the HTMLCSS, HTML4, XHTML, and MSOFFICE2K destinations all look the same (Figure 4). The 
destination that you choose depends on how you want to use it. All four destinations are compliant with Section 508 
of the U.S. Rehabilitation Act of 1973. 
 

 
Figure 4.  Output Using the HTMLCSS, HTML4, XHTML, and MSOFFICE2K Destinations 
 
 
Figure 5 compares the output from all of the destinations discussed so far. 
 
 

 
Figure 5.  Comparison of Outputs from Various Destinations 

 

In Figure 5, from the top left to the right, are IMODE, cHTML, and PHTML. From the bottom left to the right are 
HTMLCSS and HTML4/XHTML/MSOFFICE2K (which all look the same and are represented by one example). 

ONE LARGE PEPPERONI PIZZA WITH TRIPLE ANCHOVIES 
A pizza with triple anchovies means there is going to be some work after I get that pizza home; I'm going to be 
picking off anchovies right and left. If your destination is a word processor, then that means there is probably going to 
be some work to do after you create your report in SAS. 
 
Your best bet for getting your SAS output into a word processor is RTF. The ODS RTF destination does a good job of 
creating output that looks nice in your favorite word processor. ODS RTF has several options that allow customization 
to accommodate common desires and to solve problems, such as pagination and page numbering. With SAS 9.2, the 
tagsets.rtf destination offers even more options and endless customization. One of its best features is the addition of 
vertical measurement, which means pages break where you want them to, not where the word processor decides. 
 
Here is an example of using the legacy RTF destination (beginning with the release of SAS 9.2) and the new tagset-
based tagsets.rtf destination. 

 
ods rtf file="rtf.rtf"; 
 
ods tagsets.rtf file="mrtf.rtf"; 
 
proc reg data=sashelp.class; 
   model Weight = Height Age; 
run;quit;   
 
ods _all_ close; 
 

In this case, the output from both RTF destinations looks similar to the output from the HTML4 destination in Figure 6. 
 

 
Figure 6.  Output Using the Tagset-Based RTF Destination 
 

ONE LARGE PIZZA, HALF PEPPERONI, HALF BLACK OLIVE 
You might want to publish your SAS output in a paper or in a journal. Many times, the best way to do that is not by 
using a word processor, but by using LaTeX. LaTeX is a markup language that can be compiled into PDF, PostScript, 
or DeVice Independent (DVI) formats, among other things. All that you need is the free LaTeX processing system that 
you can download from the Web at www.latex-project.org/ftp.html. 
 
There are currently four LaTeX destinations. Each one has a specific niche, depending on what you desire. One 
caveat with these tagsets is that tables that are too wide for a page will be truncated. This problem is resolved in 
SAS 9.2 by leveraging the same technology that enables the RTF tagset to panel and measure text to fit on a page. 
 
There are three immediately useful LaTeX destinations: LaTeX, ColorLaTeX, and SimpleLaTeX. Both the LaTeX and 
ColorLaTeX destinations have complex style sheets and macros that allow them to create output that looks very 
much like the output from the HTML or RTF destinations. The SimpleLaTeX destination creates LaTeX with a simple 
style definition so that it can be more easily incorporated into a LaTeX document that already exists. These three 
tagsets generate LaTeX that can be directly compiled into PDF or PostScript. 

If you are writing papers or books in LaTeX, then there is another tagset that you can use called TablesOnlyLaTeX. 
This tagset allows you to create LaTeX snippets to include in another LaTeX document. 
 
Using the LaTeX destinations is a little more complicated than using some of the other destinations. The LaTeX 
destinations require an external style sheet. They require the style sheet to have a .sty extension, while the internal 
name does not. Here is an example of using the LaTeX destinations. 
 

run; 
 
/* Legacy LaTeX for ODS */ 
ods tagsets.latex file="legacy.tex"; 
 
/* Legacy LaTeX with color for ODS */ 
ods tagsets.colorlatex file="color.tex" stylesheet="sas.sty"(url="sas"); 
 
/* Simplified LaTeX output that uses plain LaTeX tables */ 
ods tagsets.simplelatex file="simple.tex" stylesheet="sas.sty"(url="sas"); 
 
/* Same as above, but only prints out tables (no titles, notes, etc.) */ 
/* Also, prints each table to a separate file */ 
ods tagsets.tablesonlylatex file="tablesonly.tex" (notop nobot) newfile=table; 
 
proc reg data=sashelp.class; 
   model Weight = Height Age; 
run;quit;   
 
ods tagsets.latex close; 
ods tagsets.colorlatex close; 
ods tagsets.simplelatex close; 
ods tagsets.tablesonlylatex close; 
 

 
When compiled into PDF, the output of these destinations looks like this (Figure 7). 
 

 
Figure 7. Output Using LaTeX Destinations 
 
 
In Figure 7, from the top left to the right, are tagsets.simplelatex and tagsets.latex. From the bottom left to the right is 
the table of contents and tagsets.colorlatex. 

ONE LARGE PIZZA, HALF PEPPERONI, HALF BLACK OLIVE—TO GO! 
If you want PDF output that is ready to go, then the ODS PDF destination is your choice. The PDF destination 
provides several options that allow you to control most of the things that you will want to control. 
 
Using the PDF destination can be as easy as using the HTML destination. 
 

ods pdf file="pdf.pdf"; 
 
proc reg data=sashelp.class; 
   model Weight = Height Age; 
run;quit;    
 
 
ods _all_ close; 

 
The output (Figure 8) looks very much like the PDF output generated by LaTeX. 
 

 
Figure 8.  Output Using the PDF Destination 

 

ONE LARGE DELUXE PIZZA TO GO, AND MAKE IT SQUARE, PLEASE 
I love a good pizza with everything on it. I don't care much for square pizzas, though. I really like that first bite on the 
pointed end of a typical slice of pizza. Suppose you need to deliver SAS reports to a group whose favorite application 
is a spreadsheet (the square pizza equivalent). For some, the CSV file format works fine to deliver reports through a 
spreadsheet. When more complex output is needed, many have resorted to using DDE (Dynamic Data Exchange). 
DDE can be very painful and slow, but almost anything can be done with it. One disadvantage of using DDE is that 
the host system must be running Microsoft Windows. Another disadvantage is that DDE can take a lot of DATA step 
code to do everything you want it to. A DDE tagset is available that removes the disadvantage of programming a 
DATA step for every data set you want to send to Microsoft Excel. (This tagset can be downloaded from the Web at 
support.sas.com/rnd/base/topics/odsmarkup/.) However, DDE is still a pain. 

The tagsets.excelxp destination is a good compromise. This destination creates XML as defined by the Microsoft 
Office XML Reference Schema (SpreadsheetML). Both Microsoft Excel (2002 and later) and the OpenOffice.org Calc 
spreadsheet program load SpreadsheetML files. Because the output file is XML, it doesn't matter what host system 
you are using. The XML can be e-mailed and used by anyone. 
 
Because this destination is a tagset, it is constantly improving and evolving. It has styles, formats, formulas, 
worksheet controls, a table of contents, a worksheet index, and an endless number of options that allow all sorts of 
controls and overrides when you need to tweak the output. 
 
When using the tagsets.excelxp destination, the most important thing to know is which version of the tagset you are 
using and where to access help. This information is displayed in a note (as shown) every time you use the 
destination. 
 

NOTE: This is the Excel XP tagset (SAS 9.1.3, v1.47, 01/04/07).   
      Add options(doc='help') to the ods statement for more information 

 
The version of the tagset changes every few months and can be downloaded from the Web at 
support.sas.com/rnd/base/topics/odsmarkup/. 
 
Using the tagsets.excelxp destination in the same way that we have used the other destinations requires you to tell it 
to embed titles and footnotes, and to put all output on one worksheet. By default, every table has its own worksheet 
and titles, and footnotes go into the page setup for printing. 
 

ods tagsets.excelxp file="Excelxp.xls"  
                    options( 
                        sheet_interval="none" 
                        embedded_titles='yes'  
                        embedded_footnotes='yes' 
                    ); 
 
proc reg data=sashelp.class; 
   model Weight = Height Age; 
run;quit;    
 
ods _all_ close; 
 

The output for this destination looks like this (Figure 9). 
 

 
Figure 9.  Output Using the tagsets.excelxp Destination 

By default, each table in the output creates a separate worksheet. This is particularly nice when using BY groups. 
Creating and naming worksheets can be controlled manually. Here is a simple example showing BY groups and the 
Printer style. Each time the first BY variable changes, a new worksheet is started. In this example, we get a workbook 
with a worksheet for every age group (Figure 10). 
 

ods tagsets.excelxp file="Excelxp2.xls" style=Printer 
                    options( 
                        sheet_interval="bygroup" 
                    ); 
 
proc sort data=sashelp.class out=work.class; 
    by Age Sex; 
     
proc print; 
   by Age Sex; 
run; 
 
ods _all_ close; 

 

 
Figure 10.  Output Using the Printer Style to Create a Workbook with Worksheets by Age Group 
 
 
The Printer style is one of the better-looking styles for tagsets.excelxp output. Journal, Statistical, BarrettsBlue, 
Normal, and Minimal styles also look nice. Some of the other styles might look good to you; that is for you to decide. 
Many styles have the same problem as the default style—the two main background colors are too closely matched, 
so Microsoft Excel makes them the same color. 
 
The tagsets.excelxp destination is just not capable of doing all the things that the other output destinations can do. 
Limitations with colors, borders, and frames keep tagsets.excelxp output from looking the same as HTML, RTF, or 
PDF. 

PIZZA THAT EVERYONE LIKES 
By now, you see that ODS delivers enough choices of output to make everyone happy. Any odd requests can be 
handled by modifying the existing tagset destinations or creating new ones. The popularity of XML as a data 
exchange format is facilitating better output that can be used by more applications. As office applications, such as 
Microsoft Office and OpenOffice.org, settle on XML standards, such as Open Document Format and Office Open 
XML, ODS will take advantage of those standards to deliver more and better output destinations. 

GLOSSARY 
 
Tagset 

Outside of SAS, a tagset is a "set" or collection of markup tags that are used to create a document that can be 
rendered using a program which understands the particular markup set of tags, or markup tagset being used. For 
example, LaTeX is a markup language tagset designed to be rendered by publishing software or printing software. 
HTML is a markup language tagset designed to create HTML pages to be rendered by a Web browser. The term 
tagset, literally means, "set of tags". 
 

Tagset Template, also known as a Tagset 
Inside SAS, a tagset template (sometimes also called a "tagset") represents a collection of instructions that tell 
SAS, through ODS, how SAS output should be "marked up" for final result file creation. So the term "tagset" 
sometimes refers to what is being written by ODS AND can also refer to how the output is being written. 

 
Destination 

The SAS Output Delivery System is a facility, within Base SAS, that allows you to route the output from your SAS 
programs and procedures to different destinations. Destinations can be categorized in two different ways: 

 
1. Categories based on what software controls rendering or viewing of the final output file 

Example categories: 
a.)   SAS controls rendering and viewing: 

LISTING destination 
OUTPUT destination 
DOCUMENT destination 
 

b.)   Third-party software controls rendering and viewing: 
HTML destination – viewed with a Web browser 
RTF destination – viewed with a word processor 
Printer family destinations: PDF, PCL, PS, Printer – rendered with the appropriate software (Acrobat 
Reader, GhostView) or device (physical printer) 
Other MARKUP destinations: 

CSV and CSVALL – viewed or rendered with spreadsheet programs 
TROFF – viewed or rendered with typesetting or publishing software 
XML – viewed or rendered with browser or custom application 
LaTeX – viewed or rendered with publishing or printing software 

 
2. Categories based in whether the output file is created with ODS MARKUP or not 

Example categories: 
a.)   Non-Markup Destinations - RTF, Printer Family, ODS OUTPUT, ODS DOCUMENT, and ODS LISTING. 

Also, ODS HTML in SAS 8 was a non-markup destination. 
b.)   Markup Destinations – all the destinations that use tagset templates to control the creation of 

destination result files. 
 

ODS MARKUP 
ODS MARKUP is the destination that is used to invoke tagset templates. Some of the ODS MARKUP tagset 
templates have had aliases assigned so that, essentially, invoking the tagset template by name allows ODS to treat 
a tagset as though it were a destination. For example, PDF is an ODS destination and when you use the ODS PDF 
invocation statement, you are said to be using the PDF destination—creating a Portable Document Format file in a 
proprietary format owned by Adobe Systems, Inc. If you design a custom tagset template to produce your flavor of 
HTML tags, using PROC TEMPLATE, you can invoke that template either like this: 
 
ODS MARKUP TAGSET=TAGSETS.MYCUSTOM FILE='custom.html'; 
 

or like this: 
 
ODS TAGSETS.MYCUSTOM FILE='custom.html'; 
 

The second invocation is still using ODS MARKUP, but essentially ODS allows you to reference 
"TAGSETS.MYCUSTOM" as though it were an ODS destination 

 
PROC TEMPLATE 

PROC TEMPLATE is the procedure that allows you to create custom tagset templates or modify existing tagset 
templates for use with the ODS MARKUP destination. 

